

Controllo del codice fiscale o della partita Iva

Silvano Mattoni - Roma

Vi invio un piccolo programma in Basic, eseguito con l'Apple II che credo possa essere adatto anche ad altri tipi di personal computer.

Si tratta di un programma per il controllo del "Codice Fiscale", valevole sia per le persone fisiche che per quelle giuridiche (o Partita IVA).

Il listato del programma, credo che sia

abbastanza semplice, ed è molto breve. Può essere usato come subroutine per la contabilità in genere e per la dichiarazione redditi ed IVA.

Il presente programma in Basic è una elaborazione del Decreto Ministeriale del 23/ 12/ 976 che consente il controllo della esatta trascrizione del Codice Fiscale, sia per le persone fisiche, che per quelle giuridiche (valevole anche per la Partita IVA).

Per le persone fisiche il Cod. Fisc. è costituito da *sedici* caratteri alfanumerici. I primi quindici indicano: il cognome, il nome, l'anno, il mese, (vedi tabella 1) il giorno aumentato di 40

per le donne ed il codice del Comune italiano o Stato estero di nascita. Il sedicesimo carattere (alfabetico) ha la funzione di controllo.

I valori numerici di detti caratteri sono indicati nella tabella 2.

Per la trasformazione di ogni carattere, nell'equivalente valore ordinale numerico, ci si può avvalere dei corrispondenti valori del codice ASCII. Infatti: ASC("A") = 65, ASC("B") = 66, ASC("C") = 67, ecc.; togliendo il valore costante di 65, al corrispondente valore ASCII di ogni carattere alfabetico si ottiene il valore numerico indicato nella citata legge.

```

100 ONERR GOTO 420
110 DATA 1,0,5,7,9,13,15,17,19,21,2,4,18,20,11,3,6,8
    ,12,14,16,10,22,25,24,23
120 DIM AL(25): FOR I = 0 TO 25: READ AL(I): NEXT
130 TEXT : HOME
140 VTAB 10: PRINT " CODICE FISCALE O PARTITA IVA ": PRINT
    : PRINT SPC( 29);
150 VTAB 12: HTAB 2: INPUT "> ";CF$
160 VTAB 20: HTAB 1: PRINT SPC( 40)
170 IF CF$ = "" THEN 250
180 A$ = "": FOR I = 1 TO LEN (CF$)
190 B$ = MID$( CF$,I,1): IF ASC (B$) < 48 THEN B$ =
    ""
200 A$ = A$ + B$: NEXT :CF$ = A$
210 IF LEN (CF$) < 14 THEN GOSUB 360:CF$ = MID$( C
    F$,1,4) + " " + MID$( CF$,5,3) + " " + MID$( CF
    $,8,3) + " " + MID$( CF$,11,1): GOTO 230
220 GOSUB 270:CF$ = MID$( CF$,1,3) + " " + MID$( CF
    $,4,3) + " " + MID$( CF$,7,5) + " " + MID$( CF$
    ,12,4) + " " + MID$( CF$,16,1)
230 IF CR$ = RIGHT$( CF$,1) THEN VTAB 20: PRINT ">
    E S A T T O . >> "CF$: GOTO 140
240 VTAB 20: PRINT CHR$( 7);CF$" E' ERRATO -> ";
    : INVERSE : PRINT CR$: NORMAL : GOTO 140
250 HOME : VTAB 23: PRINT TAB( 16)" F I N E ": END
260 REM *** CODICE FISCALE ***
270 IF LEN (CF$) < > 16 THEN VTAB 20: PRINT " LUNG
    HEZZA ERRATA ! ! " : CHR$( 7): GOTO
    140
280 CR = AL( ASC (CF$) - 65)
290 FOR I = 2 TO 14 STEP 2
300 V2 = ASC ( MID$( CF$,I,1) ) - 65: IF V2 < 0 THEN V
    2 = V2 + 17
310 V1 = ASC ( MID$( CF$,I + 1,1) ) - 65: IF V1 < 0 THEN
    V1 = V1 + 17
320 CR = CR + V2 + AL(V1)
330 NEXT :CR$ = CHR$( CR - INT ( CR / 26) * 26 + 65)
340 RETURN
350 REM *** PARTITA IVA ***
360 IF LEN (CF$) < > 11 THEN VTAB 20: PRINT " LUNG
    HEZZA ERRATA ! " : POP : GOTO 140
370 CR = 0: FOR I = 1 TO 9 STEP 2
380 V2 = 2 * VAL ( MID$( CF$,I + 1,1) ): IF V2 > 9 THEN
    V2 = V2 - 9
390 CR = CR + V2 + VAL ( MID$( CF$,I,1) )
400 NEXT :CR$ = RIGHT$( ( STR$( 100 - CR) ,1)
410 RETURN
420 VTAB 20: HTAB 1: PRINT "PRIMO CARATTERE DI COD.FI
    SC. NUMERICO": CHR$( 7)
430 GOTO 140
    
```

Tabella 1

Conversione dei mesi in un carattere alfabetico

Gennaio = A	Maggio = E	Settembre = P
Febbraio = B	Giugno = H	Ottobre = R
Marzo = C	Luglio = L	Novembre = S
Aprile = D	Agosto = M	Dicembre = T

Per i corrispondenti valori dei caratteri numerici (cioè da 0 a 9), questi differiscono di 17 posizioni nel citato codice ASCII. Infatti ASC("0") = 48, mentre ASC("A") = 65 e così di seguito. Di conseguenza, trovato il valore del carattere, mediante il codice ASCII (cioè: ASC ("—") - 65) se il risultato è negativo (= minore di 0), il carattere numerico viene trasformato nell'equivalente alfabetico aggiungendo la detta differenza di 17 posizioni.

I valori numerici ordinali dei caratteri di posizione PARI, come sopra detto, si ottengono dai corrispondenti valori del codice ASCII. Quelli di posizione DISPARI, determinato il valore ordinale, si trasformano nei corrispondenti valori numerici della citata tabella mediante un vettore dove l'indice rappresenta il numero ordinale del carattere.

Tutti i valori numerici, corrispondenti ai singoli caratteri (sia di posizione pari che di posizione dispari) vengono complessivamente addizionati. La somma si divide per 26. Si trascura la parte intera ed il resto, che rappresenta il carattere di controllo, si trasforma nel corrispondente carattere alfabetico, in codice ASCII, aggiungendo il valore numerico di 65.

Vediamo come funziona riga per riga il programma Basic.

Premesso che il primo carattere del Codice fiscale delle persone fisiche (che indica il cognome) è sempre alfabetico (ed ha posizione dispari, cioè = 1); i rimanenti quattordici caratteri possono essere presi a gruppi di due per volta (= a coppia).

Dal listato risulta alla linea:

280 CR (= carattere di controllo), si pone uguale al valore del vettore che ha per

- indice il valore ASCII - 65 del primo carattere del Codice fiscale da esaminare.
- 290 inizia un "LOOP" di sette coppie di caratteri.
- 300 V2 (= valore ASCII c.s. del carattere di posto pari); se questo valore è inferiore a zero (= carattere numerico), si aggiunge 17 per ottenere l'equivalente valore ordinale alfabetico.
- 310 V1 (= valore come sopra del carattere di posto dispari). Ugualmente si trova l'equivalente valore ordinale alfabetico.
- 320 CR (= carattere di controllo) viene incrementato (cioè sommato) della coppia dei valori esaminati (ponendo al valore di posizioni dispari = V1, il relativo vettore di trasformazione).
- 330 chiuso il ciclo (NEXT), il carattere di controllo (= CR), decurtato dei multipli di 26, viene trasformato in CHR\$ (cioè aggiungendo 65 del carattere ASCII).

Per le Persone giuridiche il Codice fiscale (equivalente, normalmente a quello della Partita Iva), è costituita da undici caratteri esclusivamente numerici. Analogamente l'ultimo carattere (= undicesimo) è un carattere numerico con la funzione di controllo.

Tabella 2

Pesi dei caratteri in posizione pari		
A o zero	= zero	N = 13
B o 1	= 1	O = 14
C o 2	= 2	P = 15
D o 3	= 3	Q = 16
E o 4	= 4	R = 17
F o 5	= 5	S = 18
G o 6	= 6	T = 19
H o 7	= 7	U = 20
I o 8	= 8	V = 21
J o 9	= 9	W = 22
K	= 10	X = 23
L	= 11	Y = 24
M	= 12	Z = 25

Pesi dei caratteri in posizione dispari		
A o zero	= 1	N = 20
B o 1	= zero	O = 11
C o 2	= 5	P = 3
D o 3	= 7	Q = 6
E o 4	= 9	R = 8
F o 5	= 13	S = 12
G o 6	= 15	T = 14
H o 7	= 17	U = 16
I o 8	= 19	V = 10
J o 9	= 21	W = 22
K	= 2	X = 25
L	= 4	Y = 24
M	= 18	Z = 23

Corrispondenza coi simboli di controllo

zero	= A	13 = N
1	= B	14 = O
2	= C	15 = P
3	= D	16 = Q
4	= E	17 = R
5	= F	18 = S
6	= G	19 = T
7	= H	20 = U
8	= I	21 = V
9	= J	22 = W
10	= K	23 = X
11	= L	24 = Y
12	= M	25 = Z

Le tre tabelle di valori che servono per calcolare il checksum del codice fiscale.

I valori numerici dei caratteri di posizione DISPARI rimangono invariati. Quelli di posizione PARI vanno raddoppiati: questi, inoltre, se costituiti da due cifre, vanno trasformati in una sola cifra (facendo la somma delle due cifre).

Tutti questi valori vanno sommati. Del risultato si prende la sola cifra relativa alle unità e si trova il complemento a dieci.

Analogamente, esaminando il listato del programma, alla linea:

360 se il codice non è di 11 caratteri è errato, salta a 140.

370 CR (= azzeramento del carattere di controllo). Inizio del "LOOP" di cinque coppie di caratteri numerici.

380 V2 (= valore numerico della cifra doppia di posto pari); se questo è maggiore di 9 (cioè se è costituita da due cifre) basta detrarre il detto 9 per ottenere il valore della somma di due cifre (Nota = prova del nove!)

390 CR (= carattere di controllo) viene incrementato (cioè sommato) con la coppia dei valori trovati.

400 chiuso il ciclo (NEXT), il carattere di controllo si trasforma in stringa (sottraendo da 100 = massimo della somma delle dieci cifre), per avere l'ultimo carattere delle unità.

Le suddette due subroutine (dal 270 al 340 e dal 360 al 410) vengono usate per esaminare il Codice fiscale o la partita IVA.

Il programma, partendo dalla linea 100, memorizza i valori dei corrispondenti caratteri alfanumerici del Codice fiscale delle persone fisiche, dimensionando il vettore e leggendo i relativi valori di tabella Ministeriale.

Alla linea:

140 chiede il Codice fiscale da controllare, battendo il solo "Return" si chiude il programma.

180/200 esamina la stringa impostata ed elimina eventuali spazi (o segni di punteggiatura che possono essere impostati per facilitare la lettura /scrittura).

210 se il Cod. Fisc. impostato è minore di 14 caratteri viene considerato relativo alle persone giuridiche (= Partita Iva) e passa alla linea 360.

220 altrimenti passa al GOSUB 270.

230 al ritorno delle subroutine esamina la corrispondenza del carattere di controllo ed eventualmente (240) visualizza il carattere determinato dalle cifre impostate poi ritorna all'inizio del programma.

250 chiude il programma. 

ERRATA CORRIGE

Nell'articolo "Riassegnamo i tasti" pubblicato su MC n. 20 nel software Apple, la figura 2 è errata!

La forma corretta è la seguente:

```

] CALL -151
*9408:85 39 A9 11 85 38 4C EA 03
*9422:EA EA EA (TOGLIE L'AUTO)
*9425:EA EA 60 (TOGLIE IL MINUS)
*9400G

```

Ricordiamo inoltre che affinché il programma funzioni non basta inserire il disassemblato di pag. 87, ma anche i dati da \$9449 a 947F, come spiegato al punto 2 della figura 1 di pag. 88.

*9400L

```

9400- A9 00 LDA ##00
9402- 85 73 STA $73
9404- A9 94 LDA ##94
9406- 85 74 STA $74
9408- 85 39 STA $39
940A- A9 11 LDA ##11
940C- 85 38 STA $38
940E- 4C EA 03 JMP $03EA

```

Disassemblato delle prime otto righe del programma Keys.Code dopo la modifica necessaria a farlo funzionare senza il minus.Code.